
1.  

Introdução à Programação 
 Português Estruturado 

 
1.1 Organização Básica de um Computador 

 

 O computador é composto basicamente de quatro unidades: unidade de 

entrada, unidade de saída, unidade de processamento(CPU) e memória.  

 

 
 
 
 

A unidade de entrada é um dispositivo utilizado para fazer a interação entre 

o usuário e o computador, de forma bem genérica, é através dela que são 

inseridos os dados os quais serão processados. A unidade de saída, por sua vez, 

serve para que o usuário veja os resultados do processamento realizado pelo 

computador. A unidade central de processamento, também conhecida como 

CPU(Central Process Unit), é responsável por todo o processamento requerido, 

transformando os dados de entrada em dados de saída. 

A CPU utiliza a memória principal (RAM) para armazenar os dados durante 

o processamento. Os dados que estão na RAM são perdidos quando o 

computador é desligado. 

 
 

 1.2 Linguagem de Máquina e Programas 
 

 As unidades componentes de um computador precisam se comunicar entre 

si. Para que haja comunicação é necessário que se estabeleça uma linguagem.

 A linguagem utilizada para comunicação entre os componentes de um 

computador é chamada de linguagem de máquina, constituída por dois símbolos 

básicos (zero e um), cada um deles denominados de bit (binary digit).  

A linguagem de máquina é um conjunto de instruções primitivas 

construida dentro de todo computador. As instruções estão na forma de código 

binário, desta forma precisa de diferentes códigos para cada instrução. Programar 

utilizando linguagem de máquina é um processo muito tedioso. Os programas são 

cdifíceis de entender e de fazer. Por exemplo uma instrução para adicionar dois 

números em código binário: 

1101101010011010 

Uma linguagem de máquina, portanto, é constituída, por uma seqüência de 

bits, ou seja, seqüências de dígitos zero e um. 

 Para que haja a comunicação do homem com o computador, é necessário 

que as palavras da linguagem escrita seja traduzida para a linguagem de 

máquina. 

 Consideramos um exemplo de uma instrução utilizando a linguagem de 

programação Assembly, considerada uma linguagem de baixo nível, porque sua 

sintaxe está mais próxima da linguagem de máquina do que da linguagem 

humana. O código abaixo utilizado para somar dois números: 

ADDF3  R1, R2, R3 

Os programas de computadores, conhecidos também como softwares, são 

instruções para o computador. Podemos dizer o que um computador deve fazer 

através de programas. Sem programas um computador é uma simples máquina 

sem função específica, talvez sirva como encosto de porta, ou algo semelhante. Os 

computadores não entendem as linguagens humanas, somos nós que precisamos 

entender e utilizar uma linguagem que seja utilizada para programá-lo e poder 

comunicar com ele. 

 As instruções escritas para um computador possuem simbolos e 

características sintáticas e semânticas que definem uma linguagem de 

programação. Uma linguagem de programação é mais legível aos seres humanos, 

pois estão muito próximas de sua lingua natural. 

 Exemplos de linguagens de programação: 

• COBOL (COmmon Business Oriented Language) 

• FORTRAN (FORmula TRANslation) 

• BASIC (Beginner All-purpose Symbolic Instructional Code) 

• Pascal (named for Blaise Pascal) 

• Ada (named for Ada Lovelace) 

• C (whose developer designed B first) 

• Visual Basic (Basic-like visual language developed by Microsoft) 

• Delphi (Pascal-like visual language developed by Borland) 

• C++ (an object-oriented language, based on C) 

• C# (a Java-like language developed by Microsoft) 

• Java 

 

 Um algoritmo escrito numa linguagem de computador é chamado de 

programa fonte ou simplesmente programa.  

 
 1.4 Lógica 

 
 A lógica é a ciência das formas do pensamento, além de estudar a correção 

do raciocínio, visto que ele é a forma mais complexa do pensamento. O homem 

por si só é um ser lógico, e essa lógica é um dom que não se aprende de uma 

hora para outra. Resulta, em geral, de experiências do dia a dia. 

 Usamos de lógica quando executamos ordenadamente uma série de rotinas 

predeterminadas, tais como: levantar da cama, pegar os chinelos, ir ao banheiro, 

tomar banho, etc. Estas ações são justificadas pela experiência prévia nestes 

procedimentos que nos fazem reconhecê-los como os mais eficazes e agradáveis 

para se começar o dia de atividade. Não existem meios de se ensinar lógica, 

podemos sim, através de um grande esforço, bastante dedicação e exercícios, 

aperfeiçoar a lógica da pessoa para que esta possa formar sua própria estrutura 

de pensamento, a qual é de fundamental importância para a arte de programar. 

 Um conceito de lógica, seria: o conjunto de leis, princípios ou métodos que 

determinam um raciocínio coerente, induzindo a uma solução prática e eficaz do 

problema. 

 
 1.5 Algoritmo 

 
 Ao contrário do que se pode pensar, o conceito de algoritmo não foi criado 

para satisfazer às necessidades da computação. Pelo contrário, a programação de 

computadores é apenas um dos campos de aplicação dos algoritmos. Na verdade, 

há inúmeros casos que podem exemplificar o uso (involuntário ou não) de 

algoritmos para a padronização do exercício de tarefas rotineiras. 

 A palavra algoritmo, à primeira vista, parece-nos estranha. Embora possua 

designação desconhecida, fazemos uso constantemente de algoritmos em nosso 

cotidiano: a maneira como uma pessoa toma banho é um algoritmo. Outros 

algoritmos freqüentemente encontrados são: 

                  • instruções para se utilizar um aparelho eletrodoméstico; 

                  • uma receita para preparo de algum prato; 

                  • guia de preenchimento para declaração do imposto de renda; 

                  • a maneira como as contas de água, luz e telefone são calculadas 

mensalmente; etc. 

 

 Um algoritmo nada mais é do que um conjunto finito de regras, bem 

definidas, para a solução de um problema em um tempo finito. 

Ex.: 

 Considerando o seguinte problema:  

“Dados três valores não nulos, A, B e C. Determinar a sua média aritmética”. As 

tarefas a serem executadas para a solução deste problema, podem ser descritas 

da seguinte forma: 

 
 

1.Obter os valores de a, b, c 
2.Calcular a média aritmética através da 
fórmula: (a+b+c)/3 
3.Comunicar o resultado obtido: média 
aritmética 
4.Terminar 

 
 

 A seqüência de tarefas acima, especificadas passo a passo, é um algoritmo, 

pois é formado por um conjunto finito de regras que podem ser consideradas bem 

definidas e cuja execução produz a solução do problema proposto, após um 

tempo finito. 

 
 
 1.6 Descrição Narrativa do Problema 
 
 Uma das maiores dificuldades encontradas pelos acadêmicos na disciplina 

de AFCA é a interpretação de texto. Não conseguir identificar aquilo que se 

pede(problema), o que se espera obter( dados de saída), o que será necessário 



AFCA - 2008 

6 | C o m p u t a ç ã o  –  U n e m a t  d e  C o l í d e r - M T 
 

para o processamento(dados de entrada) e a maneira de se processar(comandos) 

esses componentes.  

 É através do exercício da lógica, utilizando modelos de diálogo 

convencionais à lingua natural que utilizamos, que podemos construir algoritmos 

de forma narrativa para melhorar o entendimento do problema e os passos que 

levam a solução do mesmo. 

2.  

Introdução à Linguagem Algorítmica 
 Português Estruturado 

 
 
 O português estruturado apesar de não ser uma linguagem de programação 

utilizado comercialmente, ainda assim é muito eficiente no aprendizado da lógica 

para desenvolver algoritmos, consequentemente programar. 

 Por ser uma linguagem de estudo ela está muito além de ter todos os 

recursos de uma linguagem utilizada comercialmente como C, Java, Php. Mas o 

que realmente é necessário desenvolver na disciplina refere-se a capacidade de 

desenvolver algoritmos computacionais que resolvam determinados problemas. 

 O que irá ser muito comum nos problemas propostos nesta disciplina, são 

os de cunho matemático, os quais envolvem expressões aritméticas, funções, 

cálculos financeiros, etc. 

 A seguir será demonstrado de forma lógica o conteúdo proposto pela 

disciplina e que se espera que todos os acadêmicos aprovados o tenha. A prática 

com exercícios é a melhor forma de desenvolver o raciocínio e ter um melhor 

aprendizado. Bons estudos. 

 
 
2.1 Tipo de dado 
 
 Durante a execução de um programa são manipulados valores que 

precisam ser declarados a que tipo pertencem. De forma abstrata, isto existe 

porquê há uma restrição de que uma variável definida para receber um número 

não pode receber um texto, por exemplo. 

 Mas computacionalmente, isto ocorre para economizar espaço na memória. 

Como demonstrado na tabela 1, cada tipo de dado suporta um determinado 

tamanho nos valores. Este tamanho é um espaço que fica reservado na memória 

esperando por pelo valor que será preenchido posteriormente. 

Tabela 1 – Tipo de dados 

Tipo  Descrição  Valores  
Valor por 
padrão 

Inteiro valores ordinais definidos com quatro bits  
-2 147 483 648 
 2 147 483 647  

0  

Real 
Valores com parte decimal definidos com 64 
bits  

-1.7 E 308 
1.7 E 308  

0.0  

Lógico Valore lógicos - 1 bit  
verdadeiro 

falso  
falso  

Carácter Caracteres da Tabela ASCII  
ASCII(0) 

ASCII(255)  
" " (espaço)  

Texto Conjuntos de caracteres  

"Sequências de 
caracteres"  

"entre aspas"  

"" (vazio)  

 
 Como demonstrado na figura 2.1, imaginemos a memória como uma 

grande matriz que possui uma grande quantidade de espaços para serem 

utilizados. Quando declaramos um variável para um programa, durante sua 

execução esta variável é alocada em algum lugar da memória com tamanho 

correspondente ao seu tipo de dados. 

 
               
  idade          
               
  X      nome   
               
               

Figura 2.1 Variáveis em memória 
 
 
2.2 Constante 
 
 Uma informação é constante quando não sofre nenhuma alteração no 

decorrer do tempo(execução do programa). 

 Uma constante pode ser: 

� um número (inteiro ou real) 

� um valor lógico (verdadeiro ou falso) 

� um caractere (“SIM”) 

  Sintaxe: 
constante  [ tipo] [nome] <- [valor]  

constante  [ tipo] [nome] <- [expressão]  
constante  [ tipo] [nome] <- [valor] ,[nome] <- [expressão] 

  Exemplo: 
constante inteiro meses <- 12  
constante real pi <- 3.14 

2.3 Variável 

 

 Uma informação é variável quando tem a possibilidade de ser alterada em 

algum instante no decorrer do tempo(execução do programa). 

 Uma variável é uma posição de memória a qual estão associados a um tipo 

de dado e um identificador. Pelo nome dado a essa variável temos acesso a seu 

conteúdo, podendo manipulá-lo (alterar, consultar, apagar, etc). 

 
  Sintaxe:  

variavel  [ tipo] [nome] <- [expressão]  
variavel  [ tipo] [nome] <- [valor] ,[nome] <- [expressão]  
[ tipo] [nome] <- [expressão] 
[ tipo] [nome] 

Exemplo: 
variavel  inteiro idade <- 18  
variavel  real peso <- 23.14  
logico  repetente  
texto  nome <- "Antonio" , apelido <- "Sousa"  
caracter  sexo 

 

As variáveis são o elemento básico de processamento. A sua declaração 

permite definir que tipo de informação irá conter. A declaração de variáveis segue 

as seguintes regras:  

1- O nome tem de começar por uma letra ou pelo caractere  underline (_) e 

não pode ser uma palavra reservada da linguagem. 

2- O valor de inicialização tem de ser compatível com tipo de variável 

definido.  

3- Se o valor de inicialização for omitido a variável é inicializada com os 

valores por defeito  

4- É possível definir mais que uma variável utilizando o caractere virgula (,)  

5- É possível omitir a palavra variável. 

  

 
2.4 Array - Estrutura de dados homogêneas (vetor e matriz) 
 
 Define-se um conjunto de dados onde são todos do mesmo tipo e que 

podem ser acedidos através de um índice.  

Para este tipo de dados aplica-se as seguintes regras: 

1- Se a declaração omitir o termo <variável/constante> o array é variável  

2- A definição do nome obedece ás mesmas regras das variáveis. 

3- A dimensão tem de ser um valor inteiro ou uma expressão com 

resultado inteiro. 

4 - As dimensões estão contidas dentro de colchetes. 

5 - São permitidas tantas dimensões quantas as desejadas. 

6- Se o valor de inicialização for omitido todas o conjunto é inicializado com 

os valores padrão. 

7 - A inicialização é feita com os dados compreendidos entre chaves e 

separados por virgulas. 

8 - Se o número de valores for insuficiente os últimos elementos são 

iniciados com os valores padrão. 

Sintaxe: 
<variável/constante> <tipo dados> nome[dimensão]  
<variável/constante> <tipo dados> nome[dim1][dim2].  . . [dimn]  
<tipo dados> nome[dimensão]  
<tipo dados> nome[dimensão] <- {valor_1, valor_2, .  , valor_n} 

Exemplo: 
constante texto estacoes[4] <- {"inverno", "outono" , "primavera" , "verão" } 
variavel inteiro notas[3] <- {10,20}  
real salarios[12] logico bits[8]  

3.  

Operadores 
 Aritméticos, Lógicos e Relacionais 

 
3.1 Operadores Aritméticos 
 

 
3.1.1 Inteiros 

 
Estes operadores são utilizados somente para valores inteiros ou que 

utilizem somente variáveis do tipo inteiro. Os principais operadores disponíveis 
são: 

 
Operador  Significado 
 
      +       Adição  
      -      Subtração  
      /       Divisão inteira  
      *       Multiplicação  
      %    Resto da divisão inteira  
      ^     Potenciação 
Exemplo: 

 
inicio 

inteiro  i1 <- 5 , i2 <- 2 
escrever "operadores inteiros" 
escrever "\n" , i1 , " + " , i2 , " = " , i1 + i2 
escrever "\n" , i1 , " - " , i2 , " = " , i1 - i2 
escrever "\n" , i1 , " * " , i2 , " = " , i1 * i2 
escrever "\n" , i1 , " / " , i2 , " = " , i1 / i2 
escrever "\n" , i1 , " % " , i2 , " = " , i1 % i2 
escrever "\n" , i1 , " ^ " , i2 , " = " , i1 ^ i2 

fim  

 
Resultado: 
 

operadores inteiros 
5 + 2 = 7 
5 - 2 = 3 
5 * 2 = 10 
5 / 2 = 2 
5 % 2 = 1 
5 ^ 2 = 25  

 
 

3.1.2 Reais 

 
Estes operadores podem ser utilizados com dados ou variáveis do tipo real. 

Os operadores disponíveis são: 

Operador  Significado 
 
      +       Adição  
      -      Subtração  
      /       Divisão 
      *       Multiplicação  
      ^     Potenciação 
Exemplo: 

 
inicio      
real i1 <- 5.0 , i2 <- 2.0      
escrever "operadores reais"      
escrever "\n" , i1 , " + " , i2 , " = " , i1 + i2      
escrever "\n" , i1 , " - " , i2 , " = " , i1 - i2      
escrever "\n" , i1 , " * " , i2 , " = " , i1 * i2      
escrever "\n" , i1 , " / " , i2 , " = " , i1 / i2      
escrever "\n" , i1 , " ^ " , i2 , " = " , i1 ^ i2  
fim 

 
Resultado: 
 

operadores real  
5.0 + 2.0 = 7.0  
5.0 - 2.0 = 3.0  
5.0 * 2.0 = 10.0  
5.0 / 2.0 = 2.5  
5.0 ^ 2.0 = 25.0 

 
3.1.3 Texto 

 
  
Existe também um operador utilizado para concatenar letras, palavras ou frases. 
A sua sintaxe de utilização é demonstrada a seguir: 
 

Operador  Significado 
 

+      Concatenação de texto 
 

Exemplo: 
inicio 

texto nome1 <-"Linguagem" 
texto nome2 <-"Algoritmica" 
texto nome3 
nome3 <- nome1 + " " + nome2 
escrever nome3 

fim  

 
Resultado: 
 

Linguagem Algoritmica  

 
3.2 Operadores Lógicos 
 

Os operadores lógicos fornecem sempre como resultado verdadeiro ou falso. 

O operador NÃO é unário(aplica-se a uma expressão), os demais são 

binários(aplica-se a mais de uma expressão). 

Operador  Significado 

E  Disjunção  
OU     Conjunção  
XOU    Conjunção Exclusiva  
NAO    Negação 

 
Exemplo: 

inicio 
    logico l1 , l2 
    l1 <- verdadeiro 
    l2 <- falso 
    escrever "\n não " , l1 , "\t= " , nao l1 
    escrever "\n não " , l2 , "\t= " , nao l2 
 
    escrever "\n\n" , l1 , " e \t" , l2 , " = " , l 1 e l2 
    escrever "\n" , l1 , " ou \t" , l2 , " = " , l1  ou l2 
    escrever "\n" , l1 , " xou \t" , l2 , " = " , l 1 xou l2 
 
    l1 <- verdadeiro 
    l2 <- verdadeiro 
    escrever "\n\n" , l1 , " e \t" , l2 , " = " , l 1 e l2 
    escrever "\n" , l1 , " ou \t" , l2 , " = " , l1  ou l2 
    escrever "\n" , l1 , " xou \t" , l2 , " = " , l 1 xou l2 
 
    l1 <- falso 
    l2 <- falso 
    escrever "\n\n" , l1 , " e \t" , l2 , " = " , l 1 e l2 
    escrever "\n" , l1 , " ou \t" , l2 , " = " , l1  ou l2 
    escrever "\n" , l1 , " xou \t" , l2 , " = " , l 1 xou l2 

fim  

 
Resultado: 
 

não VERDADEIRO = FALSO 
não FALSO = VERDADEIRO 
 
VERDADEIRO e FALSO = FALSO 
VERDADEIRO ou FALSO = VERDADEIRO 
VERDADEIRO xou FALSO = VERDADEIRO 
 
VERDADEIRO e VERDADEIRO = VERDADEIRO 
VERDADEIRO ou VERDADEIRO = VERDADEIRO 
VERDADEIRO xou VERDADEIRO = FALSO 
 
FALSO e FALSO = FALSO 
FALSO ou FALSO = FALSO 
FALSO xou FALSO = FALSO  

 
 
3.3 Operadores Relacionais 
 

Os operadores relacionais fornecem com resultado um valor lógico 

(verdadeiro ou falso) de acordo com as seguintes definições: 

� Se os operandos forem numéricos a comparação é feita com base no seu 

valor 

� Se os operadores forem texto a comparação é feita lexicograficamente. 

� Se os operadores forem lógicos apenas estão disponíveis a igualdade e a 

diferença 

Operador  Significado 
 

=     Igual  
=/=    Diferente  
>     Maior  
>=     Maior ou igual  
<     Menor  
<=    Menor ou igual 

 
Exemplo: 

inicio 
    real l1 <- 5.0 , l2 <- 3 
    escrever "\n" , l1 , " = " , l2 , "\t= " , l1 =  l2 
    escrever "\n" , l1 , " =/= " , l2 , "\t= " , l1  =/= l2 
    escrever "\n" , l1 , " > " , l2 , "\t= " , l1 >  l2 
    escrever "\n" , l1 , " >= " , l2 , "\t= " , l1 >= l2 
    escrever "\n" , l1 , " < " , l2 , "\t= " , l1 <  l2 
    escrever "\n" , l1 , " <= " , l2 , "\t= " , l1 <= l2 

fim  

 
Resultado: 
 

5.0 = 3.0 = FALSO 
5.0 =/= 3.0 = VERDADEIRO 
5.0 > 3.0 = VERDADEIRO 
5.0 >= 3.0 = VERDADEIRO 
5.0 < 3.0 = FALSO 
5.0 <= 3.0 = FALSO  

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



AFCA - 2008 

15 | C o m p u t a ç ã o  –  U n e m a t  d e  C o l í d e r - M T 
 

4.  

Comandos de entrada e saída(E/S) 
 Ler e escrever 

 
4.1 Comando “ ler ” 
 

 

Vimos anteriormente que a maioria dos programas manipula dados que são 

fornecidos pelo usuário durante a execução do programa. Estes dados, repetindo, 

constituem a entrada do programa e, naturalmente, devem ser armazenados em 

variáveis. O comando que permite que o usuário digite no teclado algum dado 

durante a execução de um programa e que armazena este dado na memória é 

chamado de comando de entrada. É possível definir a leitura de várias variáveis 

desde que estas estejam separadas por virgulas (,). A leitura de uma variável deve 

ser precedida da sua declaração. Os valores introduzidos para cada variável 

devem ser compatíveis com o tipo definido na declaração da mesma.   

 Sintaxe: 
ler variável  
ler variavel1, variavel2 

 
Exemplo: 

inicio 
    texto nome 
    escrever "qual o seu nome:" 
    ler nome 
    escrever "bem-vindo a Unemat " , nome 

fim  

 
Resultado: 
 

qual o seu nome: aluno 
bem-vindo a Unemat aluno  

 
4.2 Comando “escrever” 
 

 Um comando de saída é utilizado quando necessitamos apresentar os 

resultados da execução do algoritmo no vídeo (monitor do computador). O 

comando escreve no console(monitor) variáveis, constantes, expressões  ou texto.  

A mesma instrução pode escrever varios elementos desde que sejam 

separados por virgulas (,). Existem três símbolos predefinidos no texto: 

\n - mudança de linha (enter) 

\t - tubulação 

\" - caracter aspas (") 

 

Sintaxe: 
escrever variável 
escrever constante 
escrever "texto" 
escrever [expressão] 
escrever variável , "texto" , expressão , . . .,  

 
Exemplo: 

inicio 
    texto nome 
    escrever "qual o seu nome:" 
    ler nome 
    escrever " \n bem-vindo  Unemat \t\" " , nome , " \" " 

fim 

 
Resultado: 
 

qual o seu nome: Aluno 
 
bem-vindo a Unemat      "Aluno"  
 

 
4.3 Comando de atribuição 
 
 O comando de atribuição (<- ) funciona como um operador que coloca 

determinado valor a uma variável. Exemplos de sua utilização podem ser vistos 

nos capítulos anteriores. 

 
4.4 Dicas para construir algoritmos 
 
 A partir deste momento temos informação suficiente para construirmos 

nossos primeiros algoritmos (programas), contudo segue algumas dicas para 

auxiliar no processo de construção: 

 a) Passo 1 : leia cuidadosamente toda a especificação do problema, faça anotações; 
 b) Passo 2 : se não entender, repita o passo 1 até entender (tire dúvidas com quem 
especificou ou conhece bem o problema); 
 c) Passo 3 : levantar todas as saídas exigidas na especificação do problema; 
 d) Passo 4 : levantar todas as entradas exigidas na especificação do problema ou 
necessárias para sua solução; 
 e) Passo 5 : verificar a necessidade de geração de valores intermediários, fazer comentários; 
 f)  Passo 6 : levantar todas as transformações necessárias para a partir da entrada 
produzir a(s) saída(s), fazer comentários; 
 g) Passo 7 : testar cada passo do algoritmo usando valores no teste, 
 h) Passo 8 : fazer uma reavaliação geral, elaborando o algoritmo 
 i) Passo 9 : executar o algoritmo. Implica em executar todas as ações descritas seguindo 
um fluxo de execução estabelecido, verificando a existência de possíveis erros no desenvolvimento 
do algoritmo. Essa atividade é conhecida como “teste de mesa”. 

5.  

Estrutura de Seleção 
 se/escolha 

  

Este tipo de estrutura permite a escolha de um caminho seqüencial a ser 

executado. Primeiramente, ela efetua um teste condicional que definirá o 

caminho de execução do algoritmo baseado no resultado deste teste. 

 Os testes condicionais são baseados na lógica convencional, podem obter o 

resultado de verdadeiro ou falso.  Estes testes podem ser efetuados sobre 

operações relacionais simples ou com expressões lógicas complexas que resultem 

em um valor lógico (verdadeiro ou falso). 

 As estruturas de seleção podem ser classificadas de 4 formas diferentes, 

sendo esta classificação baseada na organização lógica existente em cada 

situação.  Esta classificação se resume em: 

 * Seleção simples 

  * Seleção composta 

 * Seleção encadeada 

 * Seleção de múltipla escolha  

 
5.1 Estrutura Simples 
 

 Uma estrutura de seleção vai analisar se uma dada seqüência de comandos 

será ou não executada.  

 
 

Sintaxe:  
 

se (expressão lógica) então 
 Seqüência de comandos 
fimse 

Exemplo: 
inicio 

    inteiro numero 
    escrever " introduza um numero: " 
    ler numero 
    se  numero > 0 entao  
        escrever "numero positivo" 
    fimse  

fim  
 
Resultado: 
 

Introduza um numero: 2 
 
numero positivo 
 

 

 
 A seqüência de comandos só será executada se a expressão lógica for 

verdadeira, se for falsa a referida seqüência não é executada e o sistema passa a 

executar o comando seguinte ao comando SE. 

 
5.2 Estrutura Composta 
 

 
 
Sintaxe: 
 

SE (expressão lógica) ENTAO 
 Seqüência de comandos 1 
SENAO 
 Seqüência de comandos 2 
FIMSE 

 
 A seqüência de comandos 1 só será executada se a expressão lógica for 

verdadeira, se for falsa, será executada a seqüência de comandos 2. No exemplo 

abaixo o programa irá pedir que informe um número, e posteriormente ele verifica 

se ele é maior ou igual a zero, se for verdadeiro ele irá mostrar a mensagem que 

ele é positivo ou nulo. Se não for ele mostra a mensagem que o número é 

negativo. 

Exemplo: 
inicio 

    inteiro numero 
    escrever " introduza um numero: " 
    ler numero 
    se  numero >= 0 entao  
        escrever "numero positivo ou nulo" 
    senao  
        escrever "numero negativo" 
    fimse  

fim  
 
Resultado: 
 

introduza um numero: 2 
 
numero positivo ou nulo 
 

 
 
5.3 Estruturas encadeadas 

 

 
Sintaxe: 

 
SE (expressão lógica) ENTAO 
 Seqüência de comandos 1 
SENAO 
 SE (expressão lógica) ENTAO 
  Seqüência de comandos 2 
 FIMSE 
FIMSE 

 

A seqüência de comandos 1 só será executada se a expressão lógica for 

verdadeira, se for falsa, será executada a expressão lógica que está dentro do 

SENÃO. Se for verdadeira executa a sequência de comandos 2, ou se for falso, vai 

para o FIMSE e assim por diante. 

Exemplo: 
inicio 

    inteiro numero 
    escrever " introduza um numero: " 
    ler numero 
    se numero > 0 entao 
        escrever "numero positivo" 
    senao 
       se numero < 0 entao 

escrever "numero negativo" 
    fimse 

fim  

 
Resultado: 
 

introduza um numero: 2 
 
numero positivo 
 

 
 
5.4 Comando de seleção múltipla 
 

 
Este comando é utilizando quando temos múltiplas opções, e utilzar o 

comando de seleção poderia tornar-se muito confuso ou extenso. 
 

Sintaxe: 
 

Escolhe [expressão_0] 
    Caso [Expressão_1o],. . .,[Expressão_1n],: 
        [instruções] 
     . . . . . . . . . . .      
    Caso [Expressão_no],. . .,[Expressão_nn],: 
        [instruções]    
    Defeito: 
        [instruções]    

FimEscolhe  

Exemplo: 
 

inicio 
    caracter sexo 
    escrever "Qual o seu sexo(h ou m) :" 
    ler sexo 
    escrever "Olá " 
    escolhe sexo 
        caso "m" : 
            escrever "Senhor" 
        caso "f" : 
            escrever "Senhora" 
        defeito : 
            escrever "Sexo indefinido" 
    fimescolhe 
    escrever ", bemvindo ao portugol!" 

fim  
 
Resultado: 

 
Qual o seu sexo(h ou m): m 
 
Olá Senhor, bem vindo ao portugol! 

6.  

Estruturas de Repetição 
 

 
 
 O processamento de valores pode ocorrer diversas vezes sobre uma mesma 

lógica, porém os dados a serem manipulados podem ser diferentes, como no 

cálculo da tabuada de um número.  Este cálculo realiza a mesma operação de 

multiplicação sobre os valores numéricos inteiros seqüências, normalmente 

iniciados em um (1) e seguindo até o número dez (10) quando está se aprendendo 

a calcular e memorizar o resultado obtido nestas operações. 

 Sendo assim, observe a criação de um algoritmo que leia um número 

positivo e apresente o resultado de sua tabuada de um até dez (1 à 10). 

 
Exemplo: 
 Supondo que o valor desejado seja 2 então: 
 

2 x 1 = 2 
2 x 2 = 4 
2 x 3 = 6 
2 x 4 = 8 
2 x 5 = 10 
2 x 6 = 12 
2 x 7 = 14 
2 x 8 = 16 
2 x 9 = 18 
2 x 10 = 20 

 
Exemplo: 

início 
      inteiro valor 
  escrever "Informe o valor desejado: " 
      ler VALOR 
      escrever "\n" , VALOR , " x 1 = " ,  VALOR * 1   
      escrever "\n" , VALOR , " x 2 = " ,  VALOR * 2   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 3   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 4   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 5   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 6   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 7   
      escrever "\n" , VALOR , " x 3 = " ,  VALOR * 8   
      escrever "\n" , VALOR , " x 9 = " ,  VALOR * 9   
      escrever "\n" , VALOR , " x 10 = " , VALOR * 10  
fim 

Resultado: 
 

Informe o valor desejado: 3 
 

3 x 1 = 3 
3 x 2 = 6 
3 x 3 = 9 
3 x 3 = 12 
3 x 3 = 15 
3 x 3 = 18 

3 x 3 = 21 
3 x 3 = 24 
3 x 9 = 27 
3 x 10 = 30 

 

Para o exemplo acima foi desenvolvido um algoritmo que atende ao 

problema apresentado, mas se os valores tivessem que ser gerados até 1000?  

Assim o algoritmo não ficaria mais complexo, pois a operação a ser realizada seria 

a mesma, mas ele ficaria muito extenso (em número de linhas). 

 Com o intuito de controlar as manipulações necessárias sobre os dados 

armazenados, é possível utilizar instruções de repetição que permitem a repetição 

de uma operação quantas vezes forem necessárias, baseando no resultado de um 

teste condicional. 

 

6.1 Formas de Controle das Estruturas de Repetição 
 

 As estruturas de repetição possuem formas diferenciadas para controlar a 

quantidade de repetições a serem executadas.  Estes controles podem ser 

genericamente classificados em automáticos ou controlados pelo usuário do 

programa. 

• Automático: Uma variável auxiliar contará quantas vezes será executado o 

conjunto de comandos (bloco de repetição), sem interferência direta do 

usuário.  O bloco será repetido sempre a quantidade de vezes prevista pelo 

desenvolvedor do programa. 

Exemplo: Mostrar a seqüência de números de 0 a 10: 
 

Exemplo: 

inicio 
    inteiro contador 
    contador <- 0 
    enquanto  contador < 10  faz  
        escrever contador , " " 
        contador <- contador + 1 
    fimenquanto  

fim  

Resultado: 
 

0 1 2 3 4 5 6 7 8 9  

 

• Controlado pelo usuário: O programa sempre respeitará a solicitação do 

usuário, executando um bloco de repetição quantas vezes forem solicitadas 



AFCA - 2008 

24 | C o m p u t a ç ã o  –  U n e m a t  d e  C o l í d e r - M T 
 

pelo usuário, respeitando a lógica existente no programa desenvolvido.  

Exemplo: 

inicio 
      caracter resposta <- "s" 
      inteiro contador <- 0 
      real peso 
      enquanto  resposta ="s" faz  
          escrever "Informe o peso: " 
          ler peso 
          contador <- contador + 1 
           escrever "deseja continuar(s/n)? " 
           ler resposta 
      fimenquanto  
      escrever "Foram informados ", contador , " pe sos" 
fim 

  
Resultado: 
 

Informe o peso: 67 
deseja continuar(s/n)? s 
Informe o peso: 70 
deseja continuar(s/n)? n 
Foram informados 2 pesos 

 

O algoritmo anterior lê quantos pesos o usuário desejar, aplicando a mesma lógica para a 

quantidade que o usuário necessite.  Este tipo de controle da repetição também é conhecido como 

flag . 

 

6.2 Comandos de Repetição 
 

 As estruturas de repetição são subdivididas em: 

� Contada (PARA); 

� Condicional, com teste no início(ENQUANTO); 

� Condicional, com teste no final(REPETE). 

 
6.2.1 PARA ... FACA 
 
 A instrução de repetição para..faca executa um conjunto de instruções por 

uma quantidade de vezes bem definidas (conhecidas). 

Sintaxe: 
 

para  [variável numerica] de [valor inicial] ate [valor final] passo [incremento] 
    [intruções] 

próximo   

onde: 
 

• <variável_controle> é uma variável inteira (ou de um único caracter) que auxiliará na 
contagem da repetição iniciada com o valor <v_inicial> até o valor final especificado em 
<v_final> 

• <rep> é o valor do incremento dado a variável em <variável_controle>  

 O problema da tabuada, apresentado anteriormente, pode ser solucionado 

com o algoritmo que usa eficientemente a instrução de repetição para... faca. 

 
Exemplo: 

 
Início 
  inteiro valor,cont 
  escrever "Informe o valor desejado: " 
      ler VALOR 
      para  cont de  1 ate  10 passo  1 

   escrever "\n" , VALOR , " x " , cont , " = " , V ALOR * cont 
próximo     

fim 

 
Resultado: 
 

Informe o valor desejado: 2 
 
2 x 1 = 2 
2 x 2 = 4 
2 x 3 = 6 
2 x 4 = 8 
2 x 5 = 10 
2 x 6 = 12 
2 x 7 = 14 
2 x 8 = 16 
2 x 9 = 18 
2 x 10 = 20 

  
 
6.2.2 ENQUANTO .. FAZ 
 
 A instrução de repetição enquanto..faz permite a execução do conjunto de 

comandos pertencentes ao seu bloco de repetição, sendo eles executados a partir 

de um teste condicional verdadeiro.  Enquanto este teste permanecer verdadeiro 

a repetição é executada continuamente, sendo encerrada somente quando a 

condição for falsa. 

• Efetua um teste lógico antes de iniciar as instruções de repetição (ou 

looping). 

• O controle pode ser feito pelo usuário ou automaticamente por um 

contador.  

Sintaxe: 

enquanto  [condição] faz  
    [intruções] 

fimEnquanto   

 
 A seguir, será utilizado o mesmo exemplo anterior da repetição para...faca, 
porém usando a instrução de repetição enquanto...faca. 
 

Exemplo: 
 

Início 
  inteiro valor,cont 
  escrever "Informe o valor desejado: " 

      ler VALOR 
      enquanto  cont <= 10 faz  

  escrever "\n" , VALOR , " x " , cont , " = " , VA LOR * cont 
   cont <- cont + 1 

fimenquanto  
fim 

Resultado: 
 

Informe o valor desejado: 5 
 

5 x 0 = 0 
5 x 1 = 5 
5 x 2 = 10 
5 x 3 = 15 
5 x 4 = 20 
5 x 5 = 25 
5 x 6 = 30 
5 x 7 = 35 
5 x 8 = 40 
5 x 9 = 45 
5 x 10 = 50 

 
6.2.3 REPETE .. ATE 
 
 A instrução de repetição repita...ate possui características relevantes na 

sua lógica de execução. 

 As principais características dessa instrução são: 

• O teste condicional para realizar a repetição sobre o bloco de 

instruções existente entre a palavra reservada repete...ate é 

executado somente no final do bloco, conforme organização 

seqüencial do algoritmo. 

• O bloco entre as duas palavras reservadas desta instrução será 

executado no mínimo uma vez. 

• A repetição é executada somente se o teste condicional for falso, pois 

sendo ele verdadeiro a repetição é encerrada. Observe que a lógica de 

repetição desta instrução é diferente das outras duas (enquanto e 

para).  

 
Sintaxe: 

 
 repete  

    [instruções] 
 ate [condição]  

 
 Utilizando o mesmo exemplo demonstrado na instrução para...faca e 

enquanto...faca, será possível visualizar com mais facilidade a diferença no 

emprego de cada uma das instruções sobre um mesmo problema.  

 
Exemplo: 
 

Início 

  inteiro valor,cont 
  escrever "Informe o valor desejado: " 
      ler VALOR 
      repete  
   escrever "\n" , VALOR , " x " , cont , " = " , V ALOR * cont 
   cont <- cont + 1 
  ate  cont > 10 
fim 

 

Resultado: 
 

Informe o valor desejado: 5 
 

5 x 0 = 0 
5 x 1 = 5 
5 x 2 = 10 
5 x 3 = 15 
5 x 4 = 20 
5 x 5 = 25 
5 x 6 = 30 
5 x 7 = 35 
5 x 8 = 40 
5 x 9 = 45 
5 x 10 = 50 

 
 

6.3 Percentuais 

Veja esta continha simples: 

50 bolas            100  %          (50 = Todo) 

25 bolas              x    %          (25 = Parte do Todo)   

Chegaremos a conclusão: 

             X  % =  25   x    100  /  50   

Estas informações nos levam a fórmula: 

 <perc> := < Parte do Todo>   *  100  /  < Todo> 
 

Encontrando um valor final após aplicação do percentual: 

<Novo valor> := <valor anterior> * 1,<valor do percentual> 
 

Ex.: adicionar 5 %  sobre o salário e informar o valor do novo salário. 

Nsal := Salario  *  1,05; 
 

Encontrando um valor adicionado percentualmente a uma variável: 

<Novo valor> := <valor anterior>   *   0,<valor do percentual> 
 

Ex.:   adicionar 5 %  sobre o salário e informar o valor acrescido ao salário. 

Reajuste  := Salario  *  0,05 
 

 Observação: quando você não sabe o valor a ser aplicado, pois vem dentro 

de uma variável, não podemos fazer a aplicação literal do que explicamos neste 

tópico, mas precisamos traduzir em termos de programa este processo. Vamos 

fazer os mesmos exemplos anteriores, mas imaginando que o percentual virá em 

variáveis. 

a) Nsal := Salario  *  (1 + perc / 100); 

b) Nsal := Salario  *  (perc / 100); 
  

6.3 Maior Valor 

 

 Existirão momentos em que você sentirá necessidade de identificar o maior 

elemento dentro de um conjunto de dados processados. Só existe uma forma de 

sabermos qual o maior valor: comparando as grandezas do conjunto. Para 

descobrirmos qual será o maior elemento, inicializaremos uma variável, que 

armazenará o maior valor do conjunto, com um valor bem pequeno para a 

situação, e, cada dado existente no conjunto iremos comparar com esta dita 

variável. Caso o valor encontrado seja superior, este será colocado como conteúdo 

da variável maior, já que este novo dado representa a maior grandeza neste 

momento do processamento. 

 Encontrando o maior valor dentro de um conjunto: 

 Obs.: para entender melhor esse exemplo, é bom estar familiarizado com a 
estrutura de repetição comentada anteriormente. 

Exemplo: 
 

Inicio 
inteiro MaiorId <- 0, Idade,Cont 
 
  para Cont de 1 ate 10 passo 1 
       escrever "Digite sua Idade: " 
       ler Idade 
 se Idade > MaiorId entao 
         MaiorId <- Idade 
 fimse 
  proximo  
   escrever "A maior idade é: ",  MaiorId 
Fim 

Resultado: 
 

Digite sua Idade: 56 
Digite sua Idade: 45 
Digite sua Idade: 56 
Digite sua Idade: 45 
Digite sua Idade: 6 
Digite sua Idade: 4 
Digite sua Idade: 55 
Digite sua Idade: 55 
Digite sua Idade: 66 
Digite sua Idade: 7 
A maior idade é: 66 

6.3 Menor Valor 

 
 De igual forma iremos trabalhar para encontrar o elemento que possui o 

menor valor. Para tanto, iremos inicializar uma variável, que armazenará o 

menor valor do conjunto, com um valor bem alto adequado para a situação, e, 

cada dado existente no conjunto iremos comparar com esta dita variável. Caso o 

valor encontrado seja inferior, este será colocado como conteúdo da variável 

menor, já que este novo dado representa a menor grandeza neste momento do 

processamento. 

Encontrando o menor valor dentro de um conjunto: 

Exemplo: 
 

Inicio 
inteiro MenorId <- 999, Idade,Cont 
 
  para Cont de 1 ate 10 passo 1 
       escrever "Digite sua Idade: " 
       ler Idade 
 se Idade < MenorId entao 
         MenorId <- Idade 
 fimse 
  proximo  
   escrever "A menor idade é: ",  MenorId 
Fim 

 
Resultado: 
 

Digite sua Idade: 1 
Digite sua Idade: 2 
Digite sua Idade: 3 
Digite sua Idade: 4 
Digite sua Idade: 3 
Digite sua Idade: 6 
Digite sua Idade: 7 
Digite sua Idade: 8 
Digite sua Idade: 6 
Digite sua Idade: 5 
A menor idade é: 1 

7.  

Estruturas de Dados Homogêneas (vetores e matrizes) 
 

 
  As estruturas de dados consistem em organizações lógicas sobre o 

armazenamento e manipulação dos dados que serão necessários ao programa 

utilizar. Existem algumas estruturas de dados fundamentais a serem 

implementadas sobre os recursos computacionais, onde a primeira é apresentada 

a seguir. 

 Conforme abordado anteriormente, no estudo de estruturas de dados 

homogêneas na representação em algoritmo, esta estrutura permite o 

armazenamento e a manipulação de dados na memória do computador de forma 

mais organizada e eficiente no programa desenvolvido. 

 A Estrutura de Dados Homogênea Unidimensional, também chamada de 

VETOR consiste em uma única variável que pode armazenar vários valores, 

sendo todos estes valores do mesmo tipo (i.e. homogênea).  Suas principais 

características são: 

 
 * Contém vários valores (número definido); 

 * Todos valores são do mesmo tipo de dado; 

 * Possui um único nome (identificador da variável); 

 * Cada valor do conjunto é independente, de acordo com sua posição na 

estrutura.  

 
Exemplo: 
  

Suponha a existência de uma estrutura que armazena a idade de 12 
pessoas. 
 
nome do identificador = idades 
 
valores: 78 9 98 9 988 76 65 5 65 4 
índices: 0  1  2  3  4  5  6  7  8  9 

  
 Os valores correspondem aos conteúdos informados pelo usuário, ou seja, 

são as idades desejadas. 

 Os índices correspondem as posições que identificam os valores 

armazenados independentemente uns dos outros, sendo através destes índices 

realizada a manipulação especifica e independente de cada valor armazenado. 

 Suponha que seja necessário apresentar somente a idade do sexto 

indivíduo.  Como o índice da estrutura de dados sempre começa em zero, o sexto 

indivíduo possui a sua idade armazenada na 5ª posição da estrutura, sendo o 

conteúdo desta posição (5) igual a 7 (idade do sexto indivíduo armazenado na 

estrutura idades).  

 SINTAXE:  
 

  <variavel/constante> <tipo dados> nome[dimensão] 
<variavel/constante> <tipo dados> nome[dim1][dim2].  . . [dimn] 
<tipo dados> nome[dimensão] 
<tipo dados> nome[dimensão] <- {valor_1, valor_2, .  . . , valor_n}  

 
onde: 
         <variável/constante> define o tipo da informação  

<tipo de dados> é o tipo de dado que será armazenado na estrutura; 

          <nome> é o nome atribuído a estrutura (vetor); 

          <dimensão> é a quantidade de elementos que o vetor poderá armazenar. 

 
Exemplo: Receber e Apresentar as idades de 10 pessoas 
 

Inicio 
inteiro idade[10], cont 
 
  para cont de 0 ate 9 passo 1 
       escrever "Digite sua Idade: " 
       ler idade[cont] 
  proximo  
   
  para cont de 0 ate 9 passo 1 
       escrever "\n Idade: ", cont+1, " = " , idade [cont] 
  proximo 
Fim 

 
Resultado: 
 

Digite sua Idade: 1 
Digite sua Idade: 2 
Digite sua Idade: 3 
Digite sua Idade: 4 
Digite sua Idade: 5 
Digite sua Idade: 6 
Digite sua Idade: 7 
Digite sua Idade: 8 
Digite sua Idade: 9 
Digite sua Idade: 10 

 
 Idade: 1 = 1 
 Idade: 2 = 2 
 Idade: 3 = 3 
 Idade: 4 = 4 
 Idade: 5 = 5 
 Idade: 6 = 6 
 Idade: 7 = 7 
 Idade: 8 = 8 



AFCA - 2008 

32 | C o m p u t a ç ã o  –  U n e m a t  d e  C o l í d e r - M T 
 

 Idade: 9 = 9 
 Idade: 10 = 10 

  

No exemplo anterior está sendo criado um vetor chamado notas que pode 

armazenar até 10 números reais diferentes, correspondendo estes 10 números às 

notas de 10 alunos. 

A Estrutura de Dados Homogênea Multidimensional, também chamada de 

MATRIZ consiste em uma única variável que pode armazenar vários valores em 

mais de uma dimensão, sendo todos estes valores do mesmo tipo (i.e. 

homogênea). 

 
Exemplo: Receber e Apresentar as idades de 10 pessoas utilizando matrizes 
 

Inicio 
inteiro idade[2][5], cont1, cont2 
 
  para cont1 de 0 ate 1 passo 1 
      para cont2 de 0 ate 4 passo 1 
       escrever "Digite sua Idade: " 
       ler idade[cont1][cont2] 
      proximo  
  proximo  
   
 para cont1 de 0 ate 1 passo 1 
      para cont2 de 0 ate 4 passo 1 
           escrever "\n Idade: [" , cont1 , "," , c ont2 , "] = " , 

idade[cont1][cont2] 
      proximo  
  proximo  
Fim 

Resultado: 
 

Digite sua Idade: 1 
Digite sua Idade: 2 
Digite sua Idade: 3 
Digite sua Idade: 4 
Digite sua Idade: 5 
Digite sua Idade: 6 
Digite sua Idade: 7 
Digite sua Idade: 8 
Digite sua Idade: 9 
Digite sua Idade: 0 

 
 Idade: [0 , 0] = 1 
 Idade: [0 , 1] = 2 
 Idade: [0 , 2] = 3 
 Idade: [0 , 3] = 4 
 Idade: [0 , 4] = 5 
 Idade: [1 , 0] = 6 
 Idade: [1 , 1] = 7 
 Idade: [1 , 2] = 8 
 Idade: [1 , 3] = 9 
 Idade: [1 , 4] = 0 

 


